

- 1. IDENTIFICATION
- 1.1 Digital-8-12-U
- 1.2 Incremental Plotter Subroutine
- 1.3 Revised June 1, 1966

2. ABSTRACT

This subroutine moves the pen of an incremental plotter to a new position along the best straight line. The pen may be raised or lowered during the motion.

3. REQUIREMENTS

3.1 Storage

The subroutine requires one memory page (128 registers).

3.2 Equipment

Basic PDP-8
Type 350 Plotter Control and Plotter

4. USAGE

4.1 Loading

The library tape that is supplied is a symbolic tape which has no origin setting and which ends with the statement "PAUSE". This tape may be assembled with a user program (which supplies an origin) or assembled by itself if a dollar sign (\$) is supplied.

4.2 Calling Sequence

The Plot subroutine is called with an effective JMS PLOTX. The contents of the accumulator specify the operation of the subroutine:

- C(AC) = -1: The location registers internal to the subroutine are reset to 0 and the pen is raised. Control returns to the instruction following the calling JMS instruction;
- C(AC) = 0: The pen is lowered (if it was up) and is moved to the new location (see below);
- C(AC) = 1: The pen is raised (if it was down) and is moved to the new location;

The two locations following the calling JMS instructions contain, respectively, the new X coordinate and the new Y coordinate in steps (<4096). The pen is moved from the previous location to this new location along the best straight line with the pen up or down depending upon the contents of the accumulator when the subroutine is called. Control returns to the instruction following the Y coordinate.

Examples:

| <u>Initialize</u> | <u>Plot with Pen Down</u> | <u>Plot with Pen Up</u> |
|-------------------|---------------------------|-------------------------|
| CLA CMA | CLA | CLA IAC |
| JMS I PLOT | JMS I PLOT | JMS I PLOT |
| Return | X Coordinate | X Coordinate |
| | Y Coordinate | Y Coordinate |
| PLOT, PLOTX | Return | Return |
| | | |

5. RESTRICTIONS (Not Applicable)

6. DESCRIPTION

6.1 Discussion

The Plot subroutine has two registers which contain the location of the last position plotted. When the subroutine is entered, the accumulator is tested to determine if initialization is being called. If so, the location registers are set to 0, the pen is raised, and the subroutine exits. If it is not initialization, the subroutine compares the current pen position (up or down) with the requested one and raises or lowers the pen if appropriate. The new X and Y coordinates are picked up and placed in the location registers. The X and Y differences between the current location and the desired location are computed and compared. The subroutine selects motion commands depending upon the quadrant of the new location compared to the old. The possible motions are now parallel to either the X-axis (drum motion) or the Y-axis (pen motion) or a combined motion. The subroutine determines which of these motions to use, and when the new location is reached, it exits.

7. METHODS (Not Applicable)

8. FORMAT

8.1 Input Data

The X and Y coordinates are specified in numbers of steps. Increasing X corresponds to lowering the drum. Increasing Y corresponds to moving the pen left.

9. EXECUTION TIME

9.1 Minimum (Not Applicable)

9.2 Maximum (Not Applicable)

9.3 Average

This subroutine is device speed limited, although the minor subroutine, PLOTWT, could be replaced by a routine making use of the program interrupt.

10. PROGRAM

10.1 Core Map (None)

10.2 Dimension List (None)

10.3 Macro, Parameter, and Variable Lists (None)

10.4 Program Listing

```

/DIGITAL 8-12-U
/PLOT SUBROUTINE
/CALLING SEQUENCE

/   C(AC)=-1; INITIALIZE
/   C(AC)= 0; PLOT WITH PEN DOWN
/   C(AC)= 1; PLOT WITH PEN UP
/     JMS PLOTX
/     X CO-ORDINATE (IN STEPS) (RETURN IF AC=-1)
/     Y CO-ORDINATE (IN STEPS)

0200 0000 PLOTX,      0
0201 7510 SPA                /MOVE THE PEN?
0202 5220 JMP PLOTA          /NO: CONTINUE
0203 1361 TAD PLOTPN         /ADD PEN STATUS
0204 7112 CLL RTR
0205 7710 SPA CLA            /ANY CHANGE?
0206 5227 JMP PLOT1          /NO: CONTINUE
0207 7620 SNL CLA
0210 5214 JMP .+4            /LOWER THE PEN
0211 3361 DCA PLOTPN         /RAISE THE PEN
0212 6504 PLPU
0213 5216 JMP .+3
0214 2361 ISZ PLOTPN         /LOWER THE PEN
0215 6524 PLPD
0216 4370 JMS PLOTWT         /WAIT FOR FLAG
0217 5227 JMP PLOT1          /CONTINUE
0220 7200 PLOTA,      CLA
0221 6504 PLPU                /RAISE THE PEN
0222 3361 DCA PLOTPN
0223 3362 DCA PLOTNX         /0 TO X CO-ORDINATE
0224 3363 DCA PLOTNY         /0 TO Y CO-ORDINATE
0225 4370 JMS PLOTWT
0226 5600 JMP I PLOTX

/DIGITAL 8-12-U
/PAGE 2
/PICK UP ARGUMENTS

0227 1362 PLOT1,      TAD PLOTNX          /FETCH PREVIOUS X CO-ORDINATE
0230 7141 CIA CLL
0231 1600 TAD I PLOTX          /FORM NX-NPX
0232 7420 SNL                  /L=0: NX<NPX
0233 7041 CIA
0234 3364 DCA PLOTDX          /ABSOLUTE VALUE OF DIFFERENCE
0235 7004 RAL
0236 3367 DCA PLOTMV          /SAVE SIGN BIT
0237 1600 TAD I PLOTX          /SET NEW
0240 3362 DCA PLOTNX          /PREVIOUS X
0241 2200 ISZ PLOTX           /INCREMENT POINTER
0242 1363 TAD PLOTNY          /FETCH PREVIOUS Y CO-ORDINATE
0243 7141 CIA CLL
0244 1600 TAD I PLOTX          /FORM NY-NPY
0245 7420 SNL                  /<=0: NPY<NY

```

| | | | |
|------|------|-------------|---------------------------------|
| 0246 | 7041 | CIA | |
| 0247 | 3365 | DCA PLOTDY | /ABSOLUTE VALUE OF DIFFERENCE |
| 0250 | 1367 | TAD PLOTMV | /SAVE SIGN BIT |
| 0251 | 7004 | RAL | /BIT 10(1)= DRUM-DOWN(POSITIVE) |
| 0252 | 3367 | DCA PLOTMV | /BIT 11(1)=PEN-LEFT (POSITIVE) |
| 0253 | 1600 | TAD I PLOTX | /SET NEW |
| 0254 | 3363 | DCA PLOTNY | /PREVIOUS Y |
| 0255 | 2200 | ISZ PLOTX | /INCREMENT POINTER |
| 0256 | 1364 | TAD PLOTDX | |
| 0257 | 7141 | CIA CLL | |
| 0260 | 1365 | TAD PLOTDY | |
| 0261 | 7620 | SNL CLA | /L=0: DELTA Y < DELTA X |
| 0262 | 5275 | JMP PLOT2 | |
| 0263 | 1364 | TAD PLOTDX | /REVERSE NUMBERS |
| 0264 | 3366 | DCA PLOTNA | |
| 0265 | 1365 | TAD PLOTDY | |
| 0266 | 3364 | DCA PLOTDX | |
| 0267 | 1366 | TAD PLOTNA | |
| 0270 | 3365 | DCA PLOTDY | |
| 0271 | 7001 | IAC | /SET MAJOR MOTION |
| 0272 | 0367 | AND PLOTMV | /INSTRUCTION |
| 0273 | 1342 | TAD PLOTT1 | |
| 0274 | 5300 | JMP .+4 | |

/DIGITAL 8-12-U

/PAGE 3

| | | | | |
|------|------|--------------|------------|----------------------|
| 0275 | 1367 | PLOT2, | TAD PLOTMV | |
| 0276 | 7110 | CLL RAR | | |
| 0277 | 1345 | TAD PLOTT2 | | |
| 0300 | 3366 | DCA PLOTNA | | |
| 0301 | 1766 | TAD I PLOTNA | | |
| 0302 | 3340 | DCA PLOT4 | | |
| 0303 | 1367 | TAD PLOTMV | | /SET COMBINED MOTION |
| 0304 | 1350 | TAD PLOTT3 | | |
| 0305 | 3367 | DCA PLOTMV | | |
| 0306 | 1767 | TAD I PLOTMV | | |
| 0307 | 3331 | DCA PLOTDB | | |
| 0310 | 1364 | TAD PLOTDX | | |
| 0311 | 7110 | CLL RAR | | |
| 0312 | 3366 | DCA PLOTNA | | |
| 0313 | 1364 | TAD PLOTDX | | |
| 0314 | 7040 | CMA | | |
| 0315 | 3367 | DCA PLOTMV | | |
| 0316 | 2367 | PLOT3, | ISZ PLOTMV | |
| 0317 | 7410 | SKP | | |
| 0320 | 5600 | JMP I PLOTX | | /ALL DONE |
| 0321 | 1366 | TAD PLOTNA | | |
| 0322 | 1365 | TAD PLOTDY | | |
| 0323 | 3366 | DCA PLOTNA | | |
| 0324 | 1366 | TAD PLOTNA | | |
| 0325 | 7140 | CMA CLL | | |
| 0326 | 1364 | TAD PLOTDX | | |
| 0327 | 7630 | SZL CLA | | |
| 0330 | 5340 | JMP PLOT4 | | /SINGLE MOTION |
| 0331 | 0000 | PLOTDB, | 0 | /COMBINED MOTION |
| 0332 | 1364 | TAD PLOTDX | | |
| 0333 | 7041 | CIA | | |

```
0334 1366 TAD PLOTNA
0335 3366 DCA PLOTNA
0336 4370 JMS PLOTWT
0337 5316 JMP PLOT3

0340 0000 PLOT4, 0
0341 5336 JMP .-3

0342 0343 PLOTT1, .+1
0343 6511 PLPR /PEN-RIGHT
0344 6521 PLPL /PEN-LEFT
0345 0346 PLOTT2, .+1
0346 6512 PLDU /DRUM-UP
0347 6514 PLDD /DRUM-DOWN
0350 0351 PLOTT3, .+1
0351 6513 PLDU PLPR /UP-RIGHT
0352 6523 PLUD PLPL /UP-LEFT
0353 6515 PLDD PLPR /DOWN-RIGHT
0354 4355 JMS .+1 /DOWN-LEFT
0355 0000 0
0356 6514 PLDD
0357 6521 PLPL
0360 5755 JMP I .-3

/DIGITAL 8-12-U
/PAGE 4

0361 0000 PLOTPN, 0
0362 0000 PLOTNX, 0
0363 0000 PLOTNY, 0
0364 0000 PLOTDX, 0
0365 0000 PLOTDY, 0
0366 0000 PLOTNA, 0
0367 0000 PLOTMV, 0

0370 0000 PLOTWT, 0
0371 6501 PLSF /WAIT FOR DONE FLAG
0372 5371 JMP .-1 /NOT YET
0373 6502 PLCF /CLEAR FLAG
0374 5770 JMP I PLOTWT /EXIT
```

PAUSE

Digital-8-12-U

Page 6

| | |
|--------|------|
| PLOTA | 0220 |
| PLOTDB | 0331 |
| PLOTDX | 0364 |
| PLOTDY | 0365 |
| PLOTMV | 0367 |
| PLOTNA | 0366 |
| PLOTNX | 0362 |
| PLOTNY | 0363 |
| PLOTPN | 0361 |
| PLOTT1 | 0342 |
| PLOTT2 | 0345 |
| PLOTT3 | 0350 |
| PLOTWT | 0370 |
| PLOTX | 0200 |
| PLOT1 | 0227 |
| PLOT2 | 0275 |
| PLOT3 | 0316 |
| PLOT4 | 0340 |